

Algébrisation et programmation en Algol de quelques calculs pétrochimiques. Evaluation de l'erreur

par Jean-Bernard STORET

I. Introduction.

De plus en plus les pétrographes sont convaincus du rôle capital que jouent les éléments volatils, dont l'eau est le plus courant, dans la synthèse naturelle des roches. Parallèlement, les éléments traces, les anions (F^- , Cl^- , OH^- ...) sont de moins en moins négligés dans les calculs pétrochimiques maintenant utilisés universellement et rendus plus longs et plus complexes à cause du nombre élevé d'éléments analysés. Afin de débarrasser le chercheur de la tâche fastidieuse que constituent ces calculs, nous avons voulu, à notre tour, écrire un programme en langage Algol qui automatise ce travail sans risque d'erreur. Nous insistons dans ce travail sur la précision obtenue sur les résultats en fonction de l'erreur estimée sur les analyses chimiques, ce qui est à notre connaissance inédit.

Avant d'aborder la partie algébrique de notre exposé, nous allons dire quelques mots au sujet de ce que nous appellerons les « sites électroniques vacants », les « électrons disponibles », H_2O^+ , H_2O^- , et du symbolisme algébrique qu'il faudra considérer ici comme un simple ensemble d'abréviations :

— Qu'un minéral soit iso ou hétérodesmique, nous dirons qu'un élément chimique possède un site électronique vacant (électron disponible resp.) si, dans l'état de liaison (covalente ou ionique) qu'il présente ou qu'il est supposé présenter dans la structure étudiée, il complète par acceptation (par cession resp.) sa couche électronique externe.

Le nombre des sites vacants ou des électrons disponibles pour une masse donnée de roche ou de minéral est proportionnel à la valence de l'élément et, dans la suite, nous ferons abstraction de ce coefficient de proportionnalité en le prenant égal à 1. De plus, nous parlerons d'ions, même si les liaisons sont à 100 % covalentes ;

— H_2O^- représente l'eau perdue par l'échantillon de 0° à 110° C que l'on admet être l'eau d'imbibition. H_2O^+ représente le mélange gazeux de composition théorique $H_2O + O_2$ se dégageant entre 110 et 1 000° C correspondant au départ de 2 ions OH^- ;

— Quant au symbolisme, il est très simple et résumé dans le tableau suivant :

$\forall i \in [1, p]$ se lit « quelque soit » i « appartenant à » l'ensemble des valeurs 1 à p .

$\sum_{i=1}^n x_i$ se lit « somme de tous les x_i lorsque l'on donne successivement à i toutes les valeurs de 1 à n ».

$\frac{\delta F}{\delta x}$ se lit : « dérivée partielle de F par rapport à x ».

\Rightarrow se lit : ... « entraîne » ...

\ll, \gg se lisent « très inférieur à », « très supérieur à ».

\approx se lit : « voisin de ».

$x := 3x$ se lit : « la valeur $3x$ est affectée à la variable que nous appelons x ».

II. Formalisation des calculs.

Soit un échantillon dont on a analysé p cations rangés dans l'ordre de numéros 1 à p , $q - p$ anions numérotés de $p + 1$ à q (oxygène exclu car non analysé).

Nous supposons les anions monoatomiques (Cl^- , F^- ...) numérotés de $p + 1$ à $p + r$ et les anions polyatomiques (NO_3^- , SO_4^{2-} ...) numérotés de $p + r + 1$ à q .

A) Pour les cations.

Soient c_i , v_i , et m_i le pourcentage pondéral en oxyde donné par l'analyse chimique, la valence dans l'oxyde considéré, la masse atomique de l'élément (i) n° $i \forall i \in [1, p]$ contenu dans l'échantillon (roche ou minéral) étudié.

La masse moléculaire de l'oxyde est :

$$M_i = \frac{2 m_i + 16 v_i}{k_i} \quad \begin{array}{l} k_i = 1 \text{ si } v_i \text{ est impair;} \\ = 2 \text{ si } v_i \text{ est pair.} \end{array}$$

La masse moléculaire de l'oxyde ramené à 1 cation dans la molécule est :

$$M'_i = \frac{2 m_i + 16 v_i}{2} = m_i + 8 v_i$$

Le nombre de cations-grammes de (i) correspondant à c_i est : $\frac{c_i}{M'_i}$ ce qui représente $\frac{c_i}{M_i} v_i$ sites électroniques vacants ou liaisons simples.

Le nombre total de sites vacants est donc :

$$N^+ = \sum_{i=1}^p \frac{c_i}{M'_i} v_i \text{ et le nombre de cations est :}$$

$$N_o = \sum_{i=1}^p \frac{c_i}{M'_i}$$

B) Pour les anions monoatomiques (Cl^- , F^- ...) différents de $\text{O}^{=}$.

Soient v_i et m_i les grandeurs citées plus haut pour les anions (i)

$$n^\circ i \forall i \in (p + 1, p + r).$$

Soit a_i le pourcentage pondéral en élément de (i) donné par l'analyse.

La masse moléculaire M_i de l'oxyde ne peut plus être définie, mais nous désignerons par

$M'_i = m_i$ la masse moléculaire rapportée à un anion (molécule écrite avec un seul atome de (i)).

Le nombre d'anions-grammes de (i) correspondant à a_i est :

$$\frac{a_i}{M'_i}$$

ce qui représente $\frac{a_i}{M'_i} v_i$ électrons disponibles ou liaisons simples.

C) Cas des anions polyatomiques (ou cas de $\text{SO}^{=}_4$, NO^{-}_3 , OH^{-} ...).

Les ions complexes OH^- analysés sous l'appellation classique mais trompeuse de OH^{+}_2 se comportent, à l'instar de $\text{SO}^{=}_4$, NO^{-}_3 ... exactement comme les autres anions.

Soit i le n° de l'ion polyatomique ; l'on a, si $M'_i = m_i$ est la masse moléculaire de l'anion complexe de (i) :

$\frac{a_i}{M}$ anion-gramme (avec $M'_i = 17$, $v_i = 1$ pour OH^- par exemple).

L'anion (i) apporte $\frac{a_i}{M'_i} v_i$ électrons disponibles.

On peut donc maintenant évaluer le nombre total n_a d'anions-grammes analysés, distincts de l'oxygène :

$$n_a = \sum_{i=p+1}^q \frac{a_i}{M'_i}$$

Le nombre total d'électrons disponibles par les anions distincts de l'oxygène est :

$$N^- = \sum_{i=p+1}^q \frac{a_i}{M'_i} v_i$$

D) Cas de O^2 .

L'oxygène ne s'analyse pas. Il se déduit de l'hypothèse que l'équilibre électrique est vérifié. Soit N_o le nombre d'atomes-grammes d'oxygène. Il est exprimé par la formule :

$$N_o = \frac{1}{2} (N^+ - N^-) \cong 0$$

qui traduit la neutralité électrique.

E) Options de normalisation.

L'on peut définir aussi le nombre d'équivalents d'oxygène, c'est-à-dire le nombre N_{eqo} de paires de lacunes d'électrons chez les cations (cela se fait implicitement dans les calculs dans le cas fréquent où OH^- est le seul anion analysé) :

$$N_{eqo} = \frac{N^+}{2}$$

N_a nombre total d'anions (y compris l'oxygène) est souvent introduit :

$$N_a = \frac{1}{2} (N^+ - N^-) + n_a \text{ de même que } N_{io} \text{ le}$$

nombre total d'ions (\neq oxygène) :

$$N_{io} = Nc + n_a.$$

Nous désignerons désormais indifféremment les a_i et les c_i par n_i .

Tous les nombres introduits jusqu'à maintenant, c'est-à-dire :

$$\text{les } \frac{n_i}{M'_i} \text{ et } \frac{n_i}{M'_i} v_i \quad \forall i \in [1, q], N_{io}, N_a, n_a,$$

$N_{eqo}, N^+, N^-, Nc, N_o$, sont définis à la même constante de proportionnalité près. Nous allons maintenant fixer à la valeur K l'un de ces nombres que nous désignerons par M (l'on pourrait aussi introduire d'autres combinaisons linéaires de ces nombres). Ceci constitue le choix d'une option de normalisation. Il se fait en fonction de la nature de l'échantillon étudié (roche ou minéral) et de l'utilisation que l'on compte faire du résultat du calcul.

Citons par exemple :

- Calcul de la cellule de Barth : $N_o = 160$.
- Calcul des équivalents monométalliques : $N_{io} = 100$.
- Calcul de la formule structurale de la biotite : $N_a = 24$.
- Calcul de la formule structurale de la stilpnomélane :

$$\frac{n_{Si}}{M'_{Si}} = 8.$$

Le choix de la valeur numérique K , celui des nombres N que nous désignons par M étant faits,

posons : $k = \frac{K}{M}$. Les résultats x_i des calculs sont

alors les suivants :

$$(1) \left\{ \begin{array}{l} x_i = \frac{n_i}{M'_i} k \quad \forall i \in [1, q] \\ N : = Nk \quad \forall N \text{ et en particulier pour} \\ N = M \text{ on a} \\ M : = M \times \frac{K}{M} = K \end{array} \right.$$

F) Remarques.

Ces formules, qui paraissent très distinctes, fournissent dans la plupart des cas pratiques pour les valeurs traditionnelles de k des valeurs de n_i voisines.

En effet, en général pour les roches :

$$n_a \ll Nc \Rightarrow N^- \ll N^+$$

$$\text{d'où } N_a \simeq \frac{N^+}{2} \simeq N_{eqo} \simeq N_o$$

$$\text{et } Nc \simeq N_{io}.$$

De plus, BARTH a choisi $K = 160$ pour N_o de telle sorte que pour $K = 100$ relativement à N_{io} on ait des x_i voisins la plupart du temps lorsque l'on fait les calculs en utilisant ces deux options de normalisation.

Cette presque convergence des méthodes risque d'entraîner une certaine confusion dans certains esprits n'ayant pas suffisamment l'habitude des calculs pétrochimiques.

III. Calcul de la précision sur les résultats.

Nous supposons les erreurs d'analyse petites, ce qui nous permettra de négliger les variations du second ordre dans le calcul.

Différencions x_i :

$$(2) \quad dx_i = \frac{dn_i}{M'_i} k - \frac{Kn_i}{M'_i} \sum_{j=1}^q \frac{\delta M / \delta n_j}{M^2} dn_j$$

et d'après (1)

$$\begin{aligned}
 &= x_i \frac{dn_i}{n_i} - \frac{n_i}{M'_i} k \left[\frac{\delta M / \delta n_i dn_i}{M} + \sum_{\substack{j=1 \\ j \neq i}}^q \frac{\delta M / \delta n_j}{M} dn_j \right] \\
 &= x_i \left[\frac{1}{n_i} - \frac{\delta M / \delta n_i}{M} \right] dn_i + x_i \sum_{j \neq i} \frac{\delta M / \delta n_j}{M} dn_j
 \end{aligned}$$

d'où si Δx_i représente l'erreur sur le résultat, Δn_i l'erreur d'analyse et si les n_i réels deviennent les n_i mesurés :

$$(3) \quad \left| \frac{\Delta x_i}{x_i} \right| \leq \left| 1 - \frac{n_i}{M} \frac{\delta M}{\delta n_i} \right| \frac{|\Delta n_i|}{n_i} + \sum_{j \neq i} \frac{n_j}{M} \frac{|\delta M / \delta n_j|}{M} \frac{|\Delta n_j|}{n_j}$$

Cette formule est difficilement simplifiable et n'est susceptible d'être utilisée que dans le programme en Algol. Pour les calculs manuels, on peut donner une autre borne supérieure de l'erreur qui est beaucoup plus grande et peut conduire, dans certains cas, à des constatations pessimistes quant à la précision du calcul.

Reprenons le calcul et appliquons l'identité d'Euler aux formes linéaires M homogènes en n_j :

$$\sum_{j=1}^q n_j \frac{\delta M}{\delta n_j} = M$$

Après remplacement de $\frac{dn_j}{n_j}$ par sa plus grande

valeur algébrique ξ ($\xi = \sup. \frac{dn_j}{n_j} \forall j \in [1, q]$)

— ξ est donc le plus grand $\frac{dn_j}{n_j}$, de signe inconnu

en pratique, pour l'ensemble des éléments analysés on obtient :

$$dx_i < \frac{k}{M'_i} dn_i - \frac{k n_i}{M'_i} \xi \underbrace{\sum_{j=1}^q \frac{n_j \delta M / \delta n_j}{M}}_{= 1}$$

et

$$(4) \quad \left| \frac{\Delta x_i}{x_i} \right| < \frac{|\Delta n_i|}{n_i} + \sup_j \left| \frac{\Delta n_j}{n_j} \right|$$

$$\text{car } \left| \sup_j \pm \frac{\Delta n_j}{n_j} \right| = \sup_j \left| \frac{\Delta n_j}{n_j} \right|$$

Cette borne de l'erreur relative peut être très grande, car la précision relative de l'élément dosé avec le moins de précision peut être très grande (cas des éléments-traces).

Nous allons expliciter cette formule qui peut aboutir à un paradoxe si l'on veut tenir compte des erreurs dues aux éléments non analysés.

En effet, la formule (3) ne présente pas de singularité analytique si n_i mesuré = 0. Nous allons étudier la formule (4) où, pour l'élément k ($j = k$) l'on a :

$$n_k \text{ mesuré} = 0$$

Δn_k estimation de dn_k , erreur réelle est finie. On a alors

$$\sup \frac{\Delta n_j}{n_j} = \frac{\Delta n_k}{n_k} = + \infty \text{ et } \frac{\Delta x_i}{x_i} = + \infty$$

Il y a singularité analytique. Nous la lèverons de la façon suivante en considérant :

$$dn_k = n_k \text{ mesuré} - n_k \text{ réel} = - n_k \text{ réel}$$

Nous avons considéré jusqu'à maintenant que puisque $\forall j : n_j \text{ mesuré} \approx n_j \text{ réel}$ l'on pouvait partout intervertir ces deux nombres. Mais ici ce n'est plus possible car :

$$n_k \text{ mesuré} = 0.$$

Nous devons considérer la grandeur :

$$\frac{dn_k}{n_k \text{ réel}}$$

que pour $j \neq k$ on évalue par : $\frac{\Delta n_j}{n_j \text{ mesuré}}$

Nous sommes sûrs ici que : $\frac{dn_k}{n_k \text{ réel}} = -1$ et

l'on ne pourra avoir $\xi = \frac{dn_k}{n_k \text{ réel}}$ puisque cette grandeur est négative.

En pratique l'on ne tiendra donc pas compte de ces éléments.

IV. Données numériques.

La plupart des analyses chimiques trouvées dans la littérature ne précisant ni l'erreur estimée sur les résultats d'analyse, ni le procédé utilisé pour tel ou tel élément, nous avons employé, à titre indicatif, les déviations standard relatives aux analyses du granite G 1 [1]. Nous en donnons la liste, avec l'analyse :

SiO ₂	72.22 ± 0.43
TiO ₂	0.26 ± 0.067
Al ₂ O ₃	14.44 ± 0.541
Fe ₂ O ₃	0.94 ± 0.34
FeO	1.00 ± 0.135
MnO	0.03 ± 0.0122
MgO	0.39 ± 0.135
CaO	1.42 ± 0.152
Na ₂ O	3.26 ± 0.284
K ₂ O	5.51 ± 0.549

H ₂ O ⁺	0.37 ± 0.104
P ₂ O ₅	0.10 ± 0.045

L'étude faite en [1] montre également que l'erreur relative $\sigma x/x$ croît lorsque x décroît. Dans le calcul pratique effectué plus loin nous supposons que : $(\frac{\sigma x}{x}) \times x = \text{cte}$ dépendant de la nature de l'élément x .

Cela revient à dire que nous supposons, pour une méthode d'analyse donnée, l'ordre de grandeur de l'erreur absolue sur un élément constante par rapport à la teneur. Ceci nous permettra d'utiliser dans nos trois exemples les déviations standard vues plus haut.

Les trois exemples traités sont les suivants :

- Formule structurale de la stilpnomélane citée dans [2], tome 3, p. 107, analyse n° 1 ;
- Calcul des équivalents monométalliques d'une andésite citée dans [3], p. 216, analyse n° 31 ;
- Formule structurale d'un labrador cité dans [2], tome 4, p. 116, analyse n° 5.

V. Programme.

Le programme a été écrit en Algol 60 pour l'I.B.M. 7044 du laboratoire de calcul de l'Université de Grenoble. La liste en est la suivante :

5106, 0056, 0003, STORE T J.B.

EDITION ALGOL CALPE

```

'DEBUT'
1  'ENTIER' I,IPB,NPB,P,Q,J,I1 :: 'REEL' 'TABLEAU' TITRE.(1:12).,
25  OPTCLA.(1:8). ::
32  NPB:=EDONNEE ::
36  ENTREE(5,IN3,8,OPTCLA.(1).) ::
50  IN3: MODELE(''(8A6)'') ::
59  'POUR' IPB:=1 'PAS' 1 'A' NPB 'FAIRE'
68  'DEBUT' ENTREE(5,INC,12,TITRE.(1).) ::
83  IN0: MODELE(''(12A6)'') ::
93  P:=EDONNEE :: Q:=EDONNEE ::
101 'DEBUT' 'REEL' 'TABLEAU' M,MP,V,ERX,ERANA,DMI,ANA.(1:Q).,X,NOM.(1:Q+8
131 ). ::
134 'REEL' OPTION,CTE,S,NP,NC,NA,NM,NO,NEQO,NTA,NIO,N ::
159 'BOOLEEN' PRECISION ::
162 'POUR' I:=1 'PAS' 1 'A' Q 'FAIRE' ERANA.(I).:=0.0 ::
180 'POUR' I:=1 'PAS' 1 'A' Q 'FAIRE'
189 ENTREE(5,IN1,1,NOM.(I).,1,M.(I).,1,V.(I).,1,ANA.(I).,1,ERANA.(I).) ::
231 IN1: MODELE(''(A6,2X,F10.5,2X,F5.1,2X,F10.5,2X,F10.5)'') ::
246 'POUR' I:=1 'PAS' 1 'A' 8 'FAIRE' NOM.(Q+I).:=OPTCLA.(I). ::
267 CTE:=EDONNEE ::
271 ENTREE(5,IN2,1,OPTION) ::
282 IN2: MODELE(''(A6)'') ::
291 LIRE(PRECISION) ::
296
296 'COMMENT' REAJUSTEMENT DE L'ANALYSE A 100 POUR CENT ::
296 S:=0.0 ::
302 'POUR' I:=1 'PAS' 1 'A' Q 'FAIRE'
311 S:=S+ANA.(I). ::
320 'POUR' I:=1 'PAS' 1 'A' Q 'FAIRE'
329 ANA.(I).:=100.0*ANA.(I)./S ::
345
345 'COMMENT' CALCUL DES RESULTATS ::
345 'POUR' I:=1 'PAS' 1 'A' Q+8 'FAIRE' X.(I).:=0.0 ::
365 NP:=NC:=NA:=NM:=NO:=NEQO:=NTA:=NIC:=0.0 ::
385 'POUR' I:=1 'PAS' 1 'A' P 'FAIRE'
394 'DEBUT' MP.(I).:=M.(I).+8.0*V.(I). :: X.(I).:=ANA.(I)./MP.(I). ::
429 X.(Q+2).:=NC:=NC+X.(I). :: X.(Q+1).:=NP:=NP+X.(I).*V.(I).
465 'FIN' ::
467 'POUR' I:=P+1 'PAS' 1 'A' Q 'FAIRE'
478 'DEBUT' MP.(I).:=M.(I). :: X.(I).:=ANA.(I)./MP.(I). ::
504 X.(Q+3).:=NA:=NA+X.(I). :: X.(Q+4).:=NM:=NM+X.(I).*V.(I).
540 'FIN' ::
542 X.(Q+5).:=NO:=(NP-NM)/2.0 :: X.(Q+6).:=NEQO:=NP/2.0 ::
576 X.(Q+7).:=NTA:=NO+NA :: X.(Q+8).:=NIO:=NC+NA ::
602 'POUR' I:=1 'PAS' 1 'A' Q+8 'FAIRE'
613 'SI' OPTION=NOM.(I). 'ALORS'
621 'DEBUT' I1:=I :: N:=X.(I1). 'FIN' ::
634 'POUR' I:=1 'PAS' 1 'A' Q+8 'FAIRE'
645 X.(I).:=X.(I).*CTE/N ::
659
659 'COMMENT' CALCUL DE LA PRECISION DES RESULTATS ::
659 'SI' PRECISION 'ALORS'

```

```
662 'DEBUT' 'POUR' I:=1 'PAS' 1 'A' Q 'FAIRE' DMI.(I).:=0.0 ::
681 'SI' I1=Q 'INFEG' Q 'ALORS'
686 'DEBUT' DMI.(I1).:=1/MP.(I1). :: 'ALLERA' E1
700 'FIN' ::
703 'POUR' I:=1 'PAS' 1 'A' P 'FAIRE'
712 'SI' I1=Q+1 'ALORS' DMI.(I).:=V.(I)./MP.(I). 'SINON'
734 'SI' I1=Q+5 'ALORS' DMI.(I).:=V.(I)./(2.0*MP.(I).) 'SINON'
762 'SI' I1=Q+2 'OU' I1=Q+8 'ALORS' DMI.(I).:=1/MP.(I). 'SINON'
787 'SI' I1=Q+6 'ALORS' 'DEBUT' DMI.(I).:=V.(I)./(2.0*MP.(I).):='ALLERA' E1
817 'FIN' ::
820 'POUR' I:=P+1 'PAS' 1 'A' Q 'FAIRE'
831 'SI' I1=Q+5 'OU' I1=Q+7 'ALORS'
844 'DEBUT' DMI.(I).:=-V.(I)./(2.0*MP.(I).) ::
867 'SI' I1=Q+5 'ALORS' 'ALLERA' E4 'SINON' DMI.(I).:=DMI.(I).+1/MP.(I).::
894 E4: 'FIN' 'SINON'
898 'SI' I1=Q+8 'OU' I1=Q+3 'ALORS' DMI.(I).:=1/MP.(I). 'SINON'
923 'SI' I1=Q+4 'ALORS' DMI.(I).:=V.(I)./MP.(I). ::
945 E1: 'POUR' I:=1 'PAS' 1 'A' Q 'FAIRE'
956 'DEBUT' 'SI' I=I1 'ALORS'
962 'DEBUT' ERX.(I1).:=0.0 :: 'ALLERA' E2 'FIN' ::
976 ERX.(I).:=100.0*ABS(1.0-ANA.(I).*DMI.(I)./N)*(ERANA.(I)./ANA.(I).)
1014 ::
1016 'POUR' J:=1 'PAS' 1 'A' Q 'FAIRE'
1025 'DEBUT' 'SI' J=I 'ALORS' 'ALLERA' E3 'SINON'
1034 ERX.(I).:=ERX.(I).+100.0*ABS(DMI.(J).)*ERANA.(J)./N ::
1063 E3: 'FIN' ::
1067 E2: 'FIN' ::
1071 'FIN' ::
1073
1073 'COMMENT' IMPRESSIONS DES RESULTATS ::
1073 SORTIE(6,S1,12,TITRE.(1).,1,NOM.(I1).,1,CTE) ::
1098 S1:MODELE(''(1H1,12A6/1H0,13HCOMPOSANTE : ,A6,10HNORME A : ,F10.5/))'')::
1116 'POUR' I:=1 'PAS' 1 'A' Q 'FAIRE'
1125 'SI' PRECISION 'ALORS'
1128 'DEBUT'
1129 SORTIE(6,S2,1,NOM.(I).,1,X.(I).,1,ERX.(I).) ::
1157 S2: MODELE(''(1H ,10X,A6,1H:,F10.5,5H +OU-,F10.5,10H POUR CENT))'') ::
1174 'FIN'
1175 'SINON' 'SI' 'NON' PRECISION 'ALORS'
1180 SORTIE(6,S3,1,NOM.(I).,1,X.(I).) ::
1201 S3: MODELE(''(1H ,10X,A6,1H:,F10.5)'') ::
1213 SORTIE (6,S4) ::
1220 S4:MODELE (''(1H0,22H GRANDEURS CLASSIQUES:/))'') ::
1234 'POUR' I:=1 'PAS' 1 'A' 8 'FAIRE'
1243 SORTIE (6,S5,1,OPTCLA.(I).,1,X.(Q+I).) ::
1266 S5:MODELE (''(1H ,10X,A6,1H:,F10.5)'') ::
1278 'FIN' ::
1280 'FIN' ::
1282 'FIN' ::
1284 FALGOL
```

VI. Notice d'emploi. Format des données.

Les formats utilisent les notations classiques, explicitées dans [4].

A) Données d'ensemble.

Le programme est capable de traiter séquentiellement un nombre NPB (entré par EDONNEE sur la 1^{re} carte de donnée) pratiquement illimité de problèmes.

La 2^e carte, immuable, énumère en 8 A 6 les diverses grandeurs utilisées dans la partie théorique de cette note. Les équivalences de symboles sont les suivantes :

sur carte : NP équivalent à N⁺ dans le texte

»	NC	»	Nc	»
»	NA	»	<i>n_a</i>	»
»	NM	»	N ⁻	»
»	NØ	»	No	»
»	NEQØ	»	<i>Neqo</i>	»
»	NTA	»	<i>Na</i>	»
»	NIØ	»	<i>Nio</i>	»

B) Données particulières à chaque problème.

1^{re} carte : titre, références en 12 A6.

2^e carte : P : nombre de cations, Q : nombre d'ions en EDONNEE.

3^e à (Q + 3)^e carte : Nom de l'élément chimique en A 6, 2 X, masse atomique de cet élément en F 10.5, 2 X, valence de l'élément en F 5.1, 2 X, pourcentage pondéral analysé en F 10.5, 2 X, et enfin déviation standard sur l'analyse en F 10.5 si l'on désire la précision sur les résultats de calculs.

(Q + 4)^e carte : valeur de CTE appelé *k* dans le texte.

(Q + 5)^e carte : symbole NP, NC ... NIO ou SI, AL ... OH en A 6 représentent la grandeur appelée M dans le texte et que l'on norme par $M = k$.

(Q + 6)^e carte : 'VRAI' ; ou 'FAUX' ; selon que l'on désire le calcul de la précision ou non.

C) Exemple :

Données d'ensemble et données particulières pour les trois problèmes traités.

:ENTRY								
NP	NC	NA	NM	NO	NEQD	NTA	NIO	NPB
STILPNO MELANE (R.F. 1.3 P.107 NO.1)								TITRE
10::11::								P,Q
SI		28.09		4.0	48.03			
AL		26.98		3.0	6.48			
TI		47.90		4.0	0.23			
FE3		55.85		3.0	4.12			
MG		24.32		2.0	4.94			
FE2		55.85		2.0	22.88			
MN		54.94		2.0	2.67			
NA		22.991		1.0	0.0			
CA		40.08		2.0	0.83			
K		39.10		1.0	0.83			
OH		17.008		1.0	6.90			
8.0::								CTE
SI								OPTION
FAUX::								PRECISIO
ANDESITE A BIOTITE ET OXYHORNBLÉNDE (THESE V.P. P126 NO31):								TITRE
11::12::								P,Q
SI		28.09		4.0	65.65		0.43	
AL		26.98		3.0	18.45		0.541	
FE3		55.85		3.0	3.35		0.34	
FE2		55.85		2.0	0.15		0.135	
MG		24.32		2.0	0.85		0.135	
CA		40.08		2.0	2.70		0.152	
NA		22.991		1.0	4.40		0.284	
K		39.10		1.0	3.40		0.549	
TI		47.90		4.0	0.60		0.067	
P		30.975		5.0	0.40		0.045	
MN		54.94		2.0	0.05		0.0122	
OH		17.008		1.0	0.40		0.104	
100.C::								CTE
NIO								OPTION
VRAI::								PRECISIO
LABRADOR (R.F. T.4 P.116 NO.5)								
6::6::								P,Q
SI		28.09		4.0	54.28		0.43	
AL		26.98		3.0	28.28		0.541	
FE3		55.85		3.0	0.97		0.34	
CA		40.08		2.0	11.27		0.152	
NA		22.991		1.0	4.85		0.284	
K		39.10		1.0	0.41		0.549	
32.0::								CTE
NO								OPTION
VRAI::								PRECISIO
:IBSYS								

VII. Résultats numériques.

Les résultats sont imprimés de façon suffisamment claire pour se passer de commentaire :

STILPNOMELANE (R.F. T.3 P.107 NO.1)

COMPOSANTE : SI NORME A : 8.00000

SI	:	8.00000
AL	:	1.27220
TI	:	0.02881
FE3	:	0.51642
MG	:	1.22627
FE2	:	3.18720
MN	:	0.37670
NA	:	0.00000
CA	:	0.14813
K	:	0.17637
OH	:	4.06046

GRANDEURS CLASSIQUES:

NP	:	47.53407
NC	:	14.93210
NA	:	4.06046
NM	:	4.06046
NO	:	21.73681
NEQO	:	23.76704
NTA	:	25.79727
NIO	:	18.59257

ANDESITE A BIOTITE ET OXYHORNBLLENDE (THESE V.P. P126 N031):

COMPOSANTE : NIO NORME A : 100.00000

SI	:	60.05426	+0U-	3.09863	POUR CENT
AL	:	19.89335	+0U-	5.00353	POUR CENT
FE3	:	2.30612	+0U-	12.95073	POUR CENT
FE2	:	0.11476	+0U-	93.38346	POUR CENT
MG	:	1.15880	+0U-	18.80718	POUR CENT
CA	:	2.64647	+0U-	8.58384	POUR CENT
NA	:	7.80420	+0U-	8.69974	POUR CENT
K	:	3.96798	+0U-	18.15596	POUR CENT
TI	:	0.41278	+0U-	14.34964	POUR CENT
P	:	0.30979	+0U-	14.45588	POUR CENT
MN	:	0.03874	+0U-	27.70948	POUR CENT
OH	:	1.29276	+0U-	28.65993	POUR CENT

GRANDEURS CLASSIQUES:

NP	:	329.70520
NC	:	58.70724
NA	:	1.29276
NM	:	1.29276
NO	:	164.20622
NEQO	:	164.85260
NTA	:	165.49898
NIO	:	100.00000

LABRADOR (R.F. T.4 P.116 NO.5)

COMPOSANTE : NO NORME A : 32.00000

SI	:	9.83029	+OU-	1.51112	POUR CENT
AL	:	6.03682	+OU-	2.52331	POUR CENT
FE3	:	0.13220	+OU-	36.33037	POUR CENT
CA	:	2.18698	+OU-	2.85753	POUR CENT
NA	:	1.70308	+OU-	7.23982	POUR CENT
K	:	0.09473	+OU-	135.27861	POUR CENT

GRANDEURS CLASSIQUES:

NP	:	64.00000
NC	:	19.98410
NA	:	0.00000
NM	:	0.00000
NO	:	32.00000
NEQO	:	32.00000
NTA	:	32.00000
NIO	:	19.98410

23 HEURES 43 MINUTES 41 SECONDES

VIII. Conclusion.

Nous voyons que le programme vu plus haut permet d'accélérer franchement les calculs pétrochimiques (il faut à la machine un peu moins d'une seconde par calcul). Les risques d'erreur sont annulés et le calcul de la précision permet de chiffrer raisonnablement l'exactitude que l'on est en droit d'attendre des analyses chimiques pour accéder aux structures minérales ou pour émettre des raisonnements géochimiques ou autres. Notons que les résultats x_i des calculs sont toujours positifs ou nuls, et lorsqu'on lit dans le 3^e exemple que $K = 0,095 \pm 135\%$, il faut bien comprendre que l'on ne peut avoir

$$K = 0,095 - 135\% < 0.$$

Le fait que les x_i soient positifs ou nuls implique des contraintes sur les erreurs Δx_i dont nous n'avons pas tenu compte dans la théorie et qu'il est d'ailleurs difficile de prendre en considération. Nous pouvons cependant admettre que l'expression de l'erreur Δx_i sur x_i est correcte tant que les Δx_i restent petits par rapport aux x_i (ce qui n'est pas toujours le cas). Nous espérons cependant que l'utilisation de ce programme permettra aux géologues, passés les quelques heures nécessaires pour apprendre les rudiments de programmation indispensables à sa mise en œuvre, de gagner un temps précieux.

Laboratoire de Géologie,
Minéralogie et Pétrographie
de la Faculté des Sciences de Grenoble
(Institut Dolomieu).
(Laboratoire de Géologie alpine associé au C.N.R.S.)

Manuscrit déposé le 30 mars 1967.